
Welcome 

Thank you for downloading our June newsletter, we have lots of exciting 

development news to share with you in this issue. We have recently released 

version 10.05 of Spike2 with over 30 improvements and additions, which can be 

downloaded via our website. Furthermore, we are pleased to announce the 

release of our Python language interface, SonPy! This enables the conversion of 

Spike2 data to Python, see our Did you know…? section for more info. Check out 

our Spike2 section as well for information on our updated Grid views or watch 

our new tutorial video to see them in action. 

Whilst many countries are putting their plans in motion for easing lockdowns, 

many of you are still working from home. Whether you are in the lab or not, we 

are still here to help. Many users have already requested temporary copies of 

Spike2 or Signal to use at home, and we are continuing this practice for the time 

being. If you need a copy to continue your analysis, please get in touch: 

Marjorie@ced.co.uk. 

 

Training Days 

Due to the ongoing COVID-19 outbreaks, all current training events have been 

put on hold. We do however offer remote training sessions by video call either 

one-to-one or with groups. 

Join us and learn how to make the best use of Spike2 and Signal to save hours of 

repetitive analysis. Our remote sessions are free to arrange and are suitable for 

both existing and prospective users of our data acquisition and analysis systems. 

If you would like to schedule a session, please get in touch. 

If you are interested in hosting a training event in your local area once social 

distancing measures have been eased, please get in touch: Marjorie@ced.co.uk.  

 

If you see these buttons in our newsletters, it means a file or script relating to 

the section is available to download: 

 

 

Contents 

Welcome 

Training days 

Latest software 

Future meetings 

Script Spotlight  

• TextMarks editor 

 
• What is a Grid view? 

• Script – Grid view 

functions 

 

 

• How to display a 

timer 

• Script – Measure 

from states 

Did you know…? 

• SonPy – Python 

interface 

Recent questions 

• Which stimulators 

can be controlled in 

Signal? 

CED user forums 

http://ced.co.uk/downloads/latestsoftware

Data
This button signifies a Data file is available to download


Script
This button signifies a Script file is available to download


XY | Result | Other
This button signifies an XY, Result or other file is available to download

https://www.youtube.com/watch?v=V26TIJAd0tI&t

Latest versions of Spike2 and Signal 

Spike2 Released  Signal Released 

Version 10.05 05/2020  Version 7.05a 02/2020 
Version 9.11 05/2020  Version 6.05b 10/2019 
Version 8.19a 11/2019  Version 5.12a 02/2018 
Demo 03/2020  Demo 02/2020 
     

Back to contents 

Future meetings and events 

ISEK XXIII 
Virtual Meeting, 
 
July 12th – 14th 2020 

Neuroscience 2020 
Washington, DC, 
USA 
October 24th – 28th 2020 

ISAC XXI 2020 
Lisbon, 
Portugal 
November 3rd – 6th 2020 

Our meeting calendar is updated each time we receive news of cancellation or postponement due to the COVID-19 

pandemic. The full calendar is located on our website. 

Back to contents  

Script Spotlight 

TextMarks are useful event markers, holding a time stamp, a code, and a text string. With them you can annotate a 

recording using the same code for similar periods of activity, allowing subsequent automated analysis. If annotation is 

not possible during the recording you can add TextMarks afterward. We have developed this small script TextMarks 

Editor.s2s that lets you generate a range of pre-set TextMarks in a toolbar and place them with a button click. 

To operate the script first open the data file you wish to annotate. Next, run the script which will search for your 

open data file and load the toolbar:      One pre-set TextMark is already available, 

labelled “Blank”, which may be overwritten. Use Add/Remove to open a dialog to create your own pre-sets: 

 

In this dialog you should add the text to store, create a label, assign the code, and click Add to create your new pre-

set. Using Replace will overwrite the current pre-set in the drop-down box. To remove a pre-set, use the drop-down 

box to select a pre-set and click Remove. Once you have finished creating your pre-sets you are ready to begin 

annotating your data. Click OK to exit the dialog and update the toolbar.  

A cursor is placed on your data file titled “TextMark here”, drag this to where you wish to place your first TextMark. 

Alternatively, right-click the cursor and follow Cursor -> Set position to enter an exact time. Now use one of your pre-

set TextMark buttons to place it, this will create a new memory channel to hold all your annotations and place your 

first TextMark. Simply move the cursor and repeat to place where needed, using the Delete button to remove any 

mistakes. 

http://ced.co.uk/upgrades/spike2upgrade/10
http://ced.co.uk/upgrades/spike2upgrade/9
http://ced.co.uk/upgrades/spike2upgrade/8
http://ced.co.uk/upgrades/spike2demo
http://ced.co.uk/upgrades/signalupgrade/7
http://ced.co.uk/upgrades/signalupgrade/6
http://ced.co.uk/upgrades/signalupgrade/5
http://ced.co.uk/upgrades/signaldemo
https://isek.org/
https://www.sfn.org/meetings/neuroscience-2020
https://arterialchemoreceptors.org/
http://ced.co.uk/news/seeusat

'TextMarks Editor|Script for creating toolbar of TextMarks to label current data file
'===========================================================================================================
'CAMBRIDGE ELECTRONIC DESIGN LIMITED, TECHNICAL CENTRE, 139 CAMBRIDGE ROAD, MILTON, CAMBRIDGE CB24 6AX, UK
'===========================================================================================================

'Copyright © Cambridge Electronic Design.
'Author:	SG.
'v10 last modified :   12/06/2020
'SOFTWARE: This script was developed using Spike2 v10.0. It has been tested to work with Spike2 v10.00

'HEALTH WARNING:
'The script is a -work in progress- and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'-------------------------------------------------------------------------------------------------------------
Var data%;
Var mem%;
Var TxtCode%[1];
Var TxtLabel$[1]; 'Max 20 labels
Var Text$;
Var Label$:="Label", CodeV%;
Var LastHigh, LastLow;
Var TxtMks$[1];
Var Which%;
'Var n%;
Var Pos%;
Var Msg$;

Const Key$:="TextMark editor"; 'Registry entry
'Const Name$:="Contents"; 
Const CPLabel$:="TextMark here"; 'Cursor lable
Const CDLabel$:="Delete TextMark >"; 'Cursor lable

TxtMks$[0]:="BLANK"; 'First element of our TextMark list is a blank TextMark
TxtCode%[0]:=0;
TxtLabel$[0]:="BLANK";

Msg$:=Print$("Text %s Code %d", TxtMks$[0], TxtCode%[0]); 'Setup initial dialog text

data%:=frontview(); 'Get the current view handle

if viewkind(data%)<> 0 then 'Check what type of view it is 
    Message("Not a data file. Halting"); 'Fail if not a data view
    Halt;
endif

View(app(3)).Windowvisible(0); 'Hide this script

View(data%);
CursorSet(1); 'Place one cursor on the screen
CursorLabel(4, 1, CPLabel$); 'Set inital label (place TextMark)

LoadSaveRegistry%(1); 'Load any previous keys from the registry

DoToolbar();   'Produce toolbar buttons
Halt;

Func DoToolbar()    'Function start
Var i%;
ToolbarClear();     'Remove any old buttons
ToolbarSet(0, "", idle%);   'Idle routine for scrolling data when cursor moves
ToolbarSet(1, "&Quit", Quit%);    'This button returns its number
ToolBarSet(2, "ADD/REMOVE", Add%); 'Opens dialog to modify buttons
ToolbarSet(3, "&Delete", Delete%);   'Delete a TextMark
ToolbarSet(4, "Merge", Merge%); 'Import a TextMark channel to modify

If Len(TxtMks$) > 0 then 'If there are existing TextMark information
    
    For i%:=0 to Len(TxtMks$[])-1 do 'Create a toolbar button 
        ToolbarSet(5+i%, TxtLabel$[i%], PlaceTMark%);   'Link to function
    next
endif

If Mem% <= 0 then 'If memory channel does not exist
    ToolbarEnable(3, 0); 'Disable Delete
    ToolbarEnable(4, 0); 'Disable Merge
else
    ToolbarEnable(3, 1);  'Otherwise enable delete  
    ToolbarEnable(4, 1);    'and merge buttons
endif    


return Toolbar("Select...", 1023);
end;

Func idle%()   'Button 0 routine
    If ViewKind(Data%) = 0 then 'If the data view is available
        View(Data%);    'Point at it
    
    var low, high;
    
    Low:=Xlow(); High:=Xhigh(); 'Get the data time range
    
        if Low <> LastLow or High <> LastHigh then 'If there is a change in time
            Cursor(1, Low+(High-low)/2); 'Set cursor to the centre of the view
            Lastlow:=Low;   'Save the new values
            LastHigh:=high;    
        endif
    
    Endif    

if viewkind(data%)<> 0 then 'Check what type of view it is 
    Message("Data file lost. Halting"); 'Fail if not a data view
    Halt;
endif

return 1;  'This leaves toolbar active
end;

Func Add%();
Var ok%;

'n%:=Len(TxtMks$[]);

DlgCreate("TextMark Text");  'Start new dialog
DlgAllow(1023, 0, Change%);     'Function to monitor changes to dialog as they happen
'DlgString(1,"Text to add",200,"a-zA-Z0-9"); 'Text contained in TextMark
'DlgString(2,"Button Label",8,"a-zA-Z0-9"); 'Label for Toolbar button
DlgString(1,"Text to add",200); 'Text contained in TextMark
DlgString(2,"Button Label",8); 'Label for Toolbar button
DlgInteger(3,"Code",0,255,0,0,5);   'Code to add to TextMark

'If Len(TxtMks$[]) > 0 then  
    DlgList(4,"Add or remove text",TxtLabel$[]);
'Endif 

Pos%:=DlgText(Msg$, 0, 5); 'Text to change when selction changes

DlgButton(0,"Cancel");
DlgButton(1,"OK");
DlgButton(2,"Add", AddToList%); 'Adds the dialog values as a new entry
DlgButton(3, "Remove", RemoveFromList%); 'Deletes an entry
DlgButton(4, "Replace", Modify%); 'Modifies an entry
ok% := DlgShow(Text$, Label$, CodeV%, Which%);    'ok% is 0 if user cancels,  variables updated if not

If Ok% >=2 then Add%() endif
If Ok% >=0 then DoToolbar(); endif

end

Func Change%(Item%);
Var idx%;

If DlgValue(4) = 0 then 'If empty TextMark is selected
    DlgEnable(0, -3); 'Disable delete entry
else
    DlgEnable(1, -3);   'Otherwise enable delete entry
    Msg$:=Print$("Text %s Code %d", TxtMks$[DlgValue(4)], TxtCode%[DlgValue(4)]); 'Produce and...
    DlgValue$(Pos%, Msg$);  '... Update dialog text
endif

Text$ := DlgValue$(1);
Label$ := DlgValue$(2);
CodeV%:= DlgValue(3); 

If Len(Label$) < 1 then
    Label$:=Print$(CodeV%);
    DlgValue$(2, Label$);
endif

Idx%:=DlgValue(4);

If Idx% > 0 then
    If TxtMks$[Idx%] = Text$ and Label$ = TxtLabel$[Idx%] and CodeV% = TxtCode%[Idx%] then
    DlgEnable(0, -2);
else
    DlgEnable(1, -2);
    endif
endif    

If Len(TxtMks$[]) = 20+1 then 'No more than 20 TextMark entries
    DlgEnable(0, -2);
else
    DlgEnable(1, -2);
Endif    

Return 1
end

Func AddToList%();


    Resize TxtMks$[Len(TxtMks$[])+1]; 'Resize arrays to accomadate new entry
    Resize TxtLabel$[Len(TxtMks$[])];
    Resize TxtCode%[Len(TxtMks$[])];
   

    TxtMks$[Len(TxtMks$[])-1]:=DlgValue$(1); 'Place the new informaiton at the end of arrays
    TxtLabel$[Len(TxtLabel$[])-1]:=DlgValue$(2);
    TxtCode%[Len(TxtCode%[])-1]:=DlgValue(3);

Return 0
end

Func RemoveFromList%();
Var Idx%;
    Idx% := DlgValue(4); 'Get the list entry
    TxtMks$[Idx%]:="";  'Remove the entry

If Len(TxtMks$) > Idx%+1 then 'Move data in arrays to start at the deleted positon - if necessary 
    ArrConst(TxtMks$[Idx%:(Len(TxtMks$)-1)-Idx%], TxtMks$[Idx%+1:(Len(TxtMks$)-1)-Idx%]); 
    ArrConst(TxtCode%[Idx%:(Len(TxtCode%)-1)-Idx%], TxtCode%[Idx%+1:(Len(TxtCode%)-1)-Idx%]); 
    ArrConst(TxtLabel$[Idx%:(Len(TxtLabel$)-1)-Idx%], TxtLabel$[Idx%+1:(Len(TxtLabel$)-1)-Idx%]);    
Endif    
    
Resize TxtMks$[Len(TxtMks$[])-1]; 'Shorten all arrays by one
Resize TxtLabel$[Len(TxtMks$[])];
Resize TxtCode%[Len(TxtMks$[])];

Return 0
end

Func Modify%();
Var Idx%;

    Idx% := DlgValue(4); 'Get the list entry
    TxtMks$[IdX%]:=DlgValue$(1); 'Change the entries to the new values
    TxtLabel$[Idx%]:=DlgValue$(2);
    TxtCode%[Idx%]:=DlgValue(3);

Return 0
end


Func PlaceTMark%()   'Button 2 routine
Var Num%;
Num%:=ToolbarSet();

    If Mem% <= 0 then CreateMem%(); Endif 'If we do not already have a memory channel creat one
    
    Memsetitem(mem%, 0, Cursor(1), TxtCode%[Num%-5], TxtMks$[Num%-5]); 'Place a TextMark at the cursor position

return 1;  'This leaves toolbar active
end;


Func Delete%()   'Button 5 routine
CursorLabel(4,1, CDLabel$); 'Change cursor label to delete

While Interact("Move cursor to the left of the TextMark to delete", 1023, 0 , "Exit", "& Delete") = 2 do
    MemDeleteTime(mem%, 2, Cursor(1), MaxTime());
wend    

CursorLabel(4,1, CPLabel$); 
return 1;  'This leaves toolbar active
end;

Func Merge%()
Var Start, EndT;
Var ok%;
Var Chan%;

View(Data%);

DlgCreate("Merge");  'Start new dialog
DlgChan(1,"Channel to import",32);
DlgButton(0,"Cancel");
DlgButton(1,"OK");
ok% := DlgShow(Chan%);    'ok% is 0 if user cancels,  variables updated if not

If ok% = 0 then Return 1 endif 'If cancel exit

If Mem% <=0 then CreateMem%(); endif 'Create a memory channel if one does not exist  

If ChanVisible(Chan%) = 0 then Chanshow(Chan%); endif 'Show the channel to import   

CursorSet(2); 'Set two cursors
CursorLabel(4, 1, "Start"); 'Lable them
CursorLabel(4, 2, "End");

Interact("Move cursors to set the time range to import", 1023); 'Selecting an area to import

If Cursor(2) < Cursor(1) then CursorRenumber(); endif 'Reorder cursors    

Memimport(Mem%, Chan%, Cursor(1), Cursor(2));   'Import TextMarks
CursorLabel(4,1, CPLabel$); 'Reset cursor lable to Place
CursorDelete(2); 'Delete second cursor
Return 1
end


Func CreateMem%();

if CursorVisible(1) = 0 then Cursorset(1); endif 

CursorLabel(4,1,CPLabel$); 

mem% := MemChan(8, 200); 'Add new TextMark channel
ChanShow(mem%); 'Make it visible
ToolbarEnable(3,1);
ToolbarEnable(4,1);
Return 1
end

Func LoadSaveRegistry%(Type%);
Var i%;
Var Value%, Value$;
Var Num%;
Var Temp$[0];
Var Idx%;
Var Combine$;

'
If Type% = 1 then 'Load from registry
    Profile(Key$, "Num", Value%, Num%); 'Read number of entries
    
    If Value% = 0 and Num% = 0 then return 1 endif 'If no entry - exit
                                                    'Otherwise
    Resize TxtMks$[Len(TxtMks$[])+Num%];  'Extend arrays to suit
    Resize TxtLabel$[Len(TxtMks$[])];  
    Resize TxtCode%[Len(TxtMks$[])];  
    Resize Temp$[Num%+1];      
    
    For i% := 1 to Num% do  'Step through the entries
        Profile(Key$, Str$(i%), Value$, Temp$[i%]); 'Get the data
        
        If Temp$[i%] > "" then 'If the entry is not empty
            
            Idx%:=InStr(Temp$[i%], "|", 0);  'Search for Pipes to define the sectons
            TxtMks$[i%]:=Left$(Temp$[i%], Idx%-1);
            Temp$[i%]:=DelStr$(Temp$[i%], 1, Idx%);
            
            Idx%:=InStr(Temp$[i%], "|", 0);
            TxtLabel$[i%]:=Left$(Temp$[i%], Idx%-1);
            Temp$[i%]:=DelStr$(Temp$[i%], 1, Idx%);
            
            TxtCode%[i%]:=Val(Temp$[i%]);
            
         endif   
    Next  
        
    
    
else 'Write to registry
    
    Num%:=Len(TxtMks$[])-1; 'How many entries to write
    
    If num% = 0 then return 1 endif 'Exit if none       
    
    Profile(Key$, "Num", Num%); 'Set new count
    
    For i% := 1 to Num% do
        Combine$:=TxtMks$[i%]+"|"+TxtLabel$[i%]+"|"+Str$(TxtCode%[i%]); 'Create the string
        Profile(Key$, Str$(i%), Combine$); 'Write the string
    Next    
Endif    

return 1
end

Func Quit%();
Var List%[2];

    LoadSaveRegistry%(2); 'Save the current TextMark keys to registry

View(Data%); 'Data view

If Mem% > 0 then 'If we have a memort channel to save
    
    If Query("Save TextMark channel?") then 'Do you wish to save changes
        Chanlist(List%[], 128);
    
        If List%[0] > 0 then 'If there are availble disk channels ...
            Memsave(Mem%, List%[1], 0); 'Save memory chan to first available disk channel
            Chanshow(List%[1]);  'Show the new channel
            Mem%:=ChanDelete(Mem%, 0); 'Delete the memory channel
        else
            Message("No empty channels availble"); 'Error message
        endif    
    
    endif
    
endif

Halt; 'Exit the script
Return 0; 
end


TextMarks Editor.s2s
To download, right click and select Save...

It is also possible to copy existing TextMarks in your 

data file to your memory channel, should you have 

created others during recording. Click Merge and 

choose the channel you wish to copy, then use the 

cursors to select the range to copy across. When 

finished, Quit to exit the script. This will query if you 

wish to save the memory channel to the data file; if 

you do not save the channel it (and the TextMarks) 

will be lost upon closing the data file. 

 Back to contents 

 

The grid view is not a replacement for a spreadsheet, though it has some of the characteristics of one. The main 

purpose is to provide an easy way to generate tabulated data. You can copy data interactively from other sources, for 

example the Cursor Values and Regions dialogs, and manipulate it from the script language. You can move data 

between the grid and other programs that support tabulated data using a Tab character to separate columns and end 

of line characters (\n) to separate lines. 

To create a Grid view interactively, use the File menu New command or the  button in the toolbar and select Grid. 

You can either accept a default size or check the Size grid box to enter a size of your choice. Existing Grid views are 

resized by right clicking or using the View menu. There is a limit of 1,000 columns and 1,000,000 rows (version 8 had 

more restrictive limits), probably more than most people need. Grids are saved in .sgrx files using an XML format. 

The Grid can be formatted with column headers, font, alignment, type, frame, and colouring: 

Headers – The grid has one header row at the top and one header column on the left. These can be hidden or shown 

by navigating to View -> Show Column Header or Show Row Header. The column headers are changed by double 

clicking the header which opens a new dialog. The status bar at the bottom of the window (if enabled), displays the 

co-ordinates of the current grid selection as numbers. 

Font – All cells in the body of the grid use the same font. The headers use a bold version of the same font. This is 

changed through the Font menu in your toolbar. 

Row and column spacing – All rows are the same height, which is based on the font size. Columns can be sized 

individually by clicking and dragging in between cells of the column header. A standard size is set using the script 

command GrdColWidth(col%, width%), or you can optimise the columns widths to match the data displayed in 

them through the View menu. 

Alignment – You can apply Left, Centred or Right alignment on a single column basis by selecting the column and 

using the Align column function in the View menu. To apply to all columns, highlight the entire grid by clicking the top 

left most square where the headers meet and use the Align function. 

Colouring – From version 10.05 the you can change the background and text colouring of the entire grid by opening 

the colours menu. You can further change individual cell colours using the script command GrdColourSet(), or by 

right clicking an individual cell and selecting Colour Selection. 

What is a Grid view for? 

Improvements in version 10.05 – In addition to adding support for Grid colouring, we have added more script 

support for the Cursor dialogs and improved the recording of grid and cursor dialog actions to generate a script. For 

example you can now use the Script menu to Turn recording on, open a data file, open the cursor dialogs and a grid, 

select items in the cursor dialogs and paste them into the grid view, close down the data file and turn recording off. 

The resulting script will repeat your actions and serve as the basis for a data analysis procedure. See our Script 

Spotlight section for a script demonstrating these features and more.  

Watch our helpful tutorial video on Grid views help you get started. 

Back to contents 

Scripts: Spike2 

 The example script Grid View settings.s2s illustrates the use of grids from a script. It creates a toolbar with buttons: 

Quit, Settings, Change colour, and Arithmetic. The linked functions demonstrate the new and updated Grid views 

commands. To get full details of the commands and their arguments, click on a command in the script and press F1 

(or right click and use the context menu to open the online Help).  

GrdShow(cHead%, rHead%) will show/hide both the column header and the row header. We have coupled this 

command to our Settings%() function, where check boxes will interactively turn the headers on or off. 

GrdAlign(align%, col%) alters the text alignment by assigning align% with 0 (left), 1 (centred) or 2 (right). In 

our example script we alter the alignment of the entire grid by replacing col% with -2 in the Settings%() function; to 

only align selected columns use -1 instead.  

GrdSize(cols%, rows%) simply resizes the Grid view to the defined columns and rows.  

GrdColourSet() and GrdColourGet()set and return the colour values of a cell respectively. This can be done for 

both the text and background colour of cells. We have made use of GrdColourSet() in the ColourSetup%() 

function of our example script by creating a “palette” in a secondary Grid view, where the background of each cell 

uses a colour from the default palette. Using the Change colour function from the toolbar asks the user via 

Interact() to select a colour from this “palette”. GrdColourGet() obtains the red (r) blue (b) and green (g) 

colour values and applies them to your selected cells in the blank grid using GrdColourSet(). 

GrdSet() and GrdGet() are used to enter text into a cell and read text from a cell respectively. Within the 

Arithmetic%() function of the script, GrdGet() first grabs the numbers from selected cells in a column and enters 

them into an array. The script then performs a user selected calculation via the ArrSum() command, and GrdSet() 


'Grid view settings|Script example for creating and controlling a grid view
'===========================================================================================================
'CAMBRIDGE ELECTRONIC DESIGN LIMITED, TECHNICAL CENTRE, 139 CAMBRIDGE ROAD, MILTON, CAMBRIDGE CB24 6AX, UK
'===========================================================================================================

'Copyright © Cambridge Electronic Design.
'Author:	SG/LS.
'v10 last modified :   12/06/2020
'SOFTWARE: This script was developed using Spike2 v10.0. It has been tested to work with Spike2 v10.00

'HEALTH WARNING:
'The script is a -work in progress- and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'-------------------------------------------------------------------------------------------------------------

var Viewlost%;
var TxtAlign%;
var ColoursVh%;
var GrdVh%;
var Cols%:=26;
var Rows%:=52;
Var Info[0][0];
Var ColLabels$[100];

Resize info[Cols%][5];
ArrConst(Info[][4], -1);

Const Title$:= "New Grid";

view(App(3)).WindowVisible(0); 'hide running script

GetLabels();

DoToolbar();   
Halt;

Func DoToolbar()  

ToolbarClear();     
ToolbarSet(0, "Idle", Idle%);   
ToolbarSet(1, "Quit", Quit%);   
ToolbarSet(2, "New Grid", New%);   
ToolbarSet(3, "Settings", Settings%);   
ToolbarSet(4, "Change colour", Colours%);   
ToolbarSet(6, "Arithmetic", Columns%);  
ToolbarEnable(3, 0);
ToolbarEnable(4, 0);
ToolbarEnable(6, 0);
return Toolbar("Grid View Toolbar", 1023);
end;

Func Quit%()   'Button 1 routine
If ViewKind(ColoursVh%) = 17 then View(ColoursVh%).FileClose(0,-1); endif
Halt;
return 0;  'This stops the toolbar
end;

Func New%()   'Button 2 routine

GrdVh%:=FileNew(17, 3, Cols%, Rows%);   'This creates a grid view with columns and rows defined by variables
If GrdVh% < 0 then Message("Failed to create Grid view."); Return 1 endif

WindowTitle$(Print$("%s %d x %d", Title$, Cols%, Rows%));
Window(0,0,50,100);
ToolbarEnable(-1, 1);
Viewlost%:=1;

ToolbarEnable(3, 1);
ToolbarEnable(4, 1);

return 1;  'This leaves toolbar active
end;

Func Settings%()   'Button 3 routine
var ColHeadCheck%;
var RowHeadCheck%;
var ok%, dummy%;
Var Sum%;

View(GrdVh%);
Sum%:=GrdShow(-1,-1);   'This gets current status of the grid row/column headers

docase  'if Sum% has changed then 
    Case Sum% = 0 then
        ColHeadCheck%:=0;
        RowHeadCheck%:=0;
    Case Sum% = 1 then
        ColHeadCheck%:=1;
        RowHeadCheck%:=0;
    Case Sum% = 2 then
        ColHeadCheck%:=0;
        RowHeadCheck%:=1;
    Case Sum% = 3 then
        ColHeadCheck%:=1;
        RowHeadCheck%:=1;
endcase

    DlgCreate("Settings", 0, 0, 45, 8); 
    DlgLabel(1,"Define settings for Grid view", 10, 1);
    DlgCheck(2,"Show column header", 0, 2);
    DlgCheck(3,"Show row header", 25, 2);
    DlgList(4, "Text Align selected", "Left|Centred|Right", 3, 0, 3);
    DlgText("Grid Size:", 17, 4);
    DlgInteger(5,"# Columns", 1, 100, 0, 5, 1);
    DlgInteger(6,"# Rows", 1, 1000000, 0, 6, 1);
    DlgButton(0,"Cancel");
    DlgButton(1,"OK");
    DlgAllow(1023, 0, Changed%);
ok% := DlgShow(dummy%, ColHeadCheck%, RowHeadCheck%, TxtAlign%, Cols%, Rows%);    'ok% is 0 if user cancels,  variables updated if not


If ok%=1 then 
    GrdSize(Cols%, Rows%);   'resizes grid
    Resize info[Cols%][5];
    WindowSize(); 'Size view to contents
endif;

return 1;  'This leaves toolbar active
end;

func Idle%();
Var Err%;
var flag%;
Var NewPos%[4];
Var OldPos%[4];

If Viewlost% = 1 then
    flag%:=ViewKind(GrdVh%);    'If the view is lost then reset toolbar to the beginning
    
    if flag% <= 0 then
        Viewlost%:=0;
        DoToolbar();
    endif;
   Err%:=Selection(1, NewPos%[]);
If Info[NewPos%[0]][4] <> -1 then   'This section of the idle function checks the selection for the Arithmetic% function
    ArrConst(OldPos%[], Info[NewPos%[0]][0:4]);
ArrSub(OldPos%[], NewPos%[]);    
    If Arrsum(OldPos%[]) <> 0 then 'If any changes then update

    Arithmetic%(Info[NewPos%[0]][4]);
    endif
endif 
endif;

return 1
end;

func Changed%(item%);

docase
Case item% = 2 then    
        View(GrdVh%).GrdShow(DlgValue(2), -1);     'show col header
    
Case item% = 3 then 
        View(GrdVh%).GrdShow(-1, DlgValue(3));     'show row header   

Case Item% = 4 then
    TxtAlign%:=DlgValue(4);
    View(GrdVh%).GrdAlign(TxtAlign%, -1);   'Set text alignment
        
    
Case Item% = 5 or Item% = 6 then
    Cols% := DlgValue(5);  Rows% := DlgValue(6);    'Update variables of grid size
    GrdSize(Cols%, Rows%);   'resizes grid
    
endcase;

WindowSize(); 'Size view to contents
WindowTitle$(Print$("%s %d x %d", Title$, Cols%, Rows%));

return 1;
end;

func Colours%();

var ok%;
var NumCol%, NumRow%;
var r, g, b;
Var Err%;

If ViewKind(ColoursVh%) <> 17 then err%:=ColourSetup%(); endif 'Set up colour palette
If Err% < 0 then return 1 endif

While Ok% <> 1 do
FrontView(ColoursVh%);
ok%:=Interact("Select your desired colour - spacebar to accept", 1020, 0, "Return" , "& Change",  "&All");
    
    NumCol%:=MoveBy(1);  'Column
    NumRow%:=MoveBy(2);  'Row    
        GrdColourGet(0, NumCol%, NumRow%, r, g, b); 'Get selected colour
    WindowVisible(0);
    FrontView(GrdVh%);
    
    Docase  
        Case ok% = 1 then   'If user selects Return then quit out of colour select
            View(ColoursVh%);
            FileClose(0,-1);
            return 1;
        
        Case ok% = 2 then   'If user selects Change or presses spacebar, change colour of selected cells

            View(GrdVh%).GrdColourSet(0, -1, 0, r, g, b);
            
        Case ok% = 3 then   'If user selects All, change all cells
            EditSelectAll();    
            View(GrdVh%).GrdColourSet(0, -1, 0, r, g, b); 
    endcase
    
Wend    

    
return 1;
end;

func ColourSetup%();
Var LCol%, LRow%; 'local rows and columns
Var r, g, b;    'Colour values 
Var i%;         'Colour palette index

ColoursVh%:=FileNew(17, 0, 5, 8); 'Create 5x8 Grid view

If ColoursVh% < 0 then Message("Failed to create Grid view for Colour Palette."); Return ColoursVh% endif

View(ColoursVh%);
WindowTitle$("Colour palette");
WindowSize(); 'Size view to contents
WindowVisible(0);
GrdColourSet(0, -1, 0);  'Clear selection

For i% := 0 to 39 do '40 items in the colour palette
    
    ColourGet(-1, i%, r, g, b); 'Get r, g and b values for each indices
    
    LCol% := (i%/8.0);
    MoveTo(0, LCol%, LRow%); 'New selection  
    GrdColourSet(0, -1, 0, r, g, b); 'Srt the cell colour
    
    LRow% += 1;
    
    If LRow% >= 8 then
        LRow% := 0;
    endif    
    
Next

return 1;
end;

Func Columns%();
Var SCol%;
Var ok%;
Var List%;
Var ColList$[0];

Resize ColList$[GrdSize(0)];


            DlgCreate("Calc");  'Start new dialog
            DlgList(1, "Column", ColLabels$[0:GrdSize(0)-1]);
            DlgList(2,"Calculation","None|Sum|Mean|SD|Max|Min|Peak-Peak");
            DlgButton(0,"Cancel");
            DlgButton(1,"OK");
            ok% := DlgShow(SCol%, list%);    'ok% is 0 if user cancels,  variables updated if not
ToolbarText("Use mouse to highlight cells of column chosen to perform arithmetic");
If List% > 0 then
    ArrConst(Info[SCol%][4:1], List%);
  else
    ArrConst(Info[SCol%][4:1], List%-1);
Endif    

Return 1
end

Func Arithmetic%(List%);
Var Text$[0];
Var Values[0];
Var Num%;
Var i%, j%;
Var nCh%;
Var out;
Var Sum, Mean, SD;
Var Form$;
Var Result;
'Var list%;
Var OK%;
Var Pos%[4];
Var Err%;
Var MaxV, MinV;
Var Arr[2];


View(GrdVh%);
'Find selected
Err%:=Selection(1, Pos%[]);
 If err% = 0 then 
    if Pos%[0]=Pos%[2] and Pos%[1] <> Pos%[3] then
    
Resize Text$[(Pos%[3]-Pos%[1])+1];      

Num%:=GrdGet(Text$[], Pos%[0], Pos%[1]);    'Get contents of selected cells
If num% >= 1 then
    Resize Values[Num%];

For i% := 0 to Num%-1 do
    out:=Val(Text$[i%], nCh%);
    if nCh% > 0 then
        Values[j%]:= out;
        j%+=1
    else
      Resize Values[Len(Values[])-1];  
    endif    
Next
            
If Len(Values[]) = 0 then return 1; endif           
            
        docase
            case List% < 4 then
                   Sum:=Arrsum(Values[], Mean, SD);
       
                docase
                    Case List% = 0 then
                            Form$:="No Measurement";
                    Result:=0.000;
                    List%:=-1;
                        Case List% = 1 then
                            Form$:="Sum";
                            Result:=Sum;    'Holds result
                        Case List% = 2 then
                            Form$:="Mean";
                            Result:=Mean;   'Holds result
                        Case List% = 3 then
                            Form$:="SD";
                            Result:=SD; 'Holds result
                    endcase   
                
            case List% = 4 then
                Form$:="MAX";
                MaxV:=Max(Values[]);
                Result:=Values[MaxV];   'Holds result
            case List% = 5 then
                Form$:="MIN";
                MinV:=Min(Values[]);
                Result:=Values[MinV];   'Holds result
            case List% = 6 then
                Form$:="Pk-PK";
                MinV:=Min(Values[]);
                MaxV:=Max(Values[]);
                Arr[0]:=Values[MinV];
                Arr[1]:=Values[MaxV];
                ArrDiff(Arr[]);
                Result:=Arr[1]; 'Holds result
        endcase    
            ArrConst(Info[Pos%[0]][0:4], Pos%[0:4]);
            ArrConst(Info[Pos%[0]][4:1], List%);
                GrdSet(Print$("%s %f",Form$, Result), Pos%[0], -1); 'Change column header with the result
                
            endif
  endif
endif

Return 1
end

Proc GetLabels();
Var Temp%;
Temp%:=FileNew(17, 0, Len(ColLabels$[]), 1);
View(Temp%);
GrdGet(trans(ColLabels$[]), 0, -1); 'Get current column header labels

FileClose(0, -1);

return
end

Grid View settings.s2s
To download, right click and select Save...

https://www.youtube.com/watch?v=V26TIJAd0tI&t

then overwrites the column header with the answer. There is an important caveat when dealing with numerical data 

in a Grid view; all data within the Grid is stored as a string of text. Therefore, any numbers entered into the grid must 

be converted back to real or integer data after using GrdGet(); much of the Arithmetic%() function deals with this, 

as well as converting the calculated output back to a string before inputting with GrdSet(). 

MoveTo() is used to select cells in a Grid view, and MoveBy() will select cells in the grid relative to the last position. 

The Selection() command is used to obtain information on selected cells, i.e. the number of selections or the 

positions of a selection. 

Back to contents 

 

Information windows are ideal for displaying a timer. Info windows are either created with the View menu, Info 

windows -> Create New Window command or from a script with the InfoNew() function. This is only available when a 

data or XY view is the current view. Info windows are mostly used to display user-defined information in a large font 

for visibility from across the lab, with options for a timer to be configured. They could also be used to display your 

protocol steps at the same time. Simply typing out the protocol is an option, or info windows can also display an 

image file instead of the standard background; resizing the window will cause the contents to automatically scale to 

fill the available space. To further improve the amount of visible area, either the area containing the buttons or the 

window title area (or both) can be hidden if desired. These and other options are available by right-clicking on the 

info window. 

To go into more detail, each info window contains a 

user-defined text string, the optional background 

image, and configurable colours. The text string uses 

special fields which are replaced by the timer value or 

other data extracted from the associated view and 

updated at user-defined intervals. Meaning for 

example, if you wished to display the most recent Y 

value of an XY plot in giant red letters which updated 

with every sweep of data, you could configure that. 

Info windows are attached to a file, memory or XY 

view and are saved as a resource; when the associated view is re-opened any 

associated info windows will be re-created. Furthermore, up to 10 Info windows 

can be attached to each data view, and by using the available script functions one 

can create detailed displays from a script. 

This configuration demonstrates two info windows in use. The first info window is 

linked to the data file and shows the current frame, maximum frames obtained, 

the current state, and a timer in seconds. The text string is written as Frame: %f 

(Max %m)\nCurrent State: %s\nTimer: %t (s), where special fields are prefixed 

with %, and \n denotes a new line. The second info window is linked to an XY trend plot of the current frame vs the 

maximum amplitude and shows the current frames maximum and the previous frames maximum. The text string is 

Maximum: %ry1 (V)\nLast maximum: %py1 (V).  

Back to contents 

 

 

Is it possible to display a timer on screen?

 


 
	 
		 
		 
		
		 
			 
				 
			
		
		 
		 
		 
			 
				 
				 
				 
				 
				 
				 
				 
				 
			
		
		 
		 
		 
			 
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
			
			 
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
					 
				
				 
					 
				
			
			 
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
				
				 
					 
					 
					 
				
				 
					 
				
			
		
		 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
			 
		
		 
		 
			 
				 
					 
						 
						 
					
					 
						 
						 
					
				
				 
					 
						 
						 
					
					 
						 
						 
					
				
				 
					 
						 
						 
					
					 
						 
						 
					
				
				 
					 
						 
						 
					
					 
						 
						 
					
				
			
		
		 
		 
		 
	
	 
		 
			 
			 
			 
			 
			 
				 
			
			 
				 
			
			 
				 
				 
			
			 
			 
			 
		
		 
			 
				 
			
			 
			 
			 
				 
					 
					 
					 
						 
						 
							 
						
						 
					
					 
					 
						 
					
				
			
		
	
	 
		 
	


Info Windows Example.sgcx
To download, right click and select Save...

Scripts: Signal 

Our Measurefromstates.sgs script will plot a selected measurement from an evoked response in an XY view. The 

script is specifically written to obtain measurements per state from a file containing multiple frame states. The 

measurements available are: curve area, mean, slope, area, sum, modulus, maximum, minimum, peak to peak, RMS 

amplitude, standard deviation, extreme, peaks, troughs, point count, SEM, and RMS error. 

Essentially this script helpfully reduces the number of steps and button clicks needed to produce an XY trend plot of 

the simple measurements above. Being more visual than the XY trend plot dialog, and by automatically creating an XY 

channel per state, the script is far quicker than attempting to produce the plots manually. Upon running the script 

you are presented with a dialog that allows you to choose the channel, the measurement, and how many states to 

perform the measurement on. The script then uses the Interact() function to give you the chance to mark out the 

evoked response, before proceeding to the produce the XY trend plot.  

The script is available to download from our website here. A suitable example data file, MEP example.cfs, is included 

in the .zip file for you to try out the scripts function. 

Back to contents 

Did you know…? 

Alongside the release of Spike2 update 10.05 we have created SonPy, our Python language interface. With the 

interface, anyone with a knowledge of Python may access the data in Spike2 data files or create entirely new data 

files. It is possible to create and edit all Spike2 channels: waveform, events, and markers. 

The interface is available to download from our website, with a full help guide included with the installation. We have 

also included a few test scripts to demonstrate creating a new 32- or 64-bit file, reading an existing WaveForm or 

RealWave channel, and reading all data within a Spike2 data file. 

We are interested to see our users run with this, so please let us know your comments and feedback. We hope the 

interface is of use to you! 

Back to contents 

Recent Questions 

What peripheral stimulators can I trigger using Signal? 

Any stimulator that has an input for digital triggering 

can be triggered by Signal. First configure a square TTL 

pulse in your pulse outputs of appropriate length 

(usually 1ms is sufficient but consult your stimulators 

handbook) and connect the correct Digital Output of 

your 1401 to the stimulator and you are set.  

However, do note that in this type of setup full control 

of the stimulus intensity is given by the stimulator, it is 

only possible to control the timing of the stimulus in 

Signal. 

In addition, the Digitimer DS5 can be controlled via the 1401s DAC output. Much like the digital triggered stimulators, 

this is set up in Signal via the pulse outputs, however instead a pulse being configure on the Digital Outputs, a 

waveform should be configured on a DAC output. The analogue voltage input on the DS5 is translated into an isolated 

constant current stimulus (up to ±50mA), replicating the shape of the input waveform. 

http://ced.co.uk/downloads/scriptsiganal
http://ced.co.uk/upgrades/spike2sonpy

The Digitimer DS8 on the other hand can be software 

controlled through Signal to set the stimulus parameters. 

Much like the Signal controlled Transcranial Magnetic 

Stimulation (TMS) devices, the DS8 is added through the 

Auxiliary devices menu when Multiple Frame States is 

enabled. There all the stimulus parameters may be 

configured, as well as the timing of the stimulus being set 

via the pulse outputs.  

 

 

Back to contents 

CED User forums 

Try the CED Forums bulletin board for software and hardware support. 

If you have any comments about the newsletter format and content, please get in touch: Marjorie@ced.co.uk. 

To adjust your subscription preferences, please visit our website: www.ced.co.uk/upgrades/subscribeenews. 

Technical Centre, 139 Cambridge Road, 

Telephone: (01223) 420186 

